
Abstract

We describe mGTK, a Standard ML language binding
for the Gtk+ toolkit. Gtk+ is a graphical toolkit for the X
Window System, and provides an object-oriented C lan-
guage API. Since Standard ML is a mostly-functional
language without object types, constructing a binding to
Gtk+ is not a trivial task. In mGTK, a single-inheritance
class hierarchy is encoded using SML’s type system.
Most of the mGTK binding is machine generated, to best
utilize the limited manpower of the project.

The goal of the mGTK project is “just” to present a
type-safe interface to Gtk+ for SML programmers. This
contrasts with GUI libraries for functional languages,
which concentrate on producing good user interfaces:
there are several SML graphical user interface libraries
available for this task. With mGTK, SML applications
have access to the mature, complete and familiar Gtk+
user interface.

1 Introduction
A good Standard ML (SML) binding to Gtk+ is an ad-
vantage for both the SML community and for the Gtk+
community. The SML community benefits in a couple
of ways. First, SML programmers get access to a good
general-purpose graphical user interface (GUI) library
with a large range of modern widgets: the SML com-
munity is sorely missing such a library. Second, it is
a step in providing full access to the whole GNOME
platform. This will make it possible for SML program-
mers to make real applications without having to invent
add-hoc solutions to many standard problems, such as
database access. There are also a couple of advantages
of an SML binding to the Gtk+ community. First, such a
binding can help open up the small, but important, mar-
ket of teaching languages. Second, as SML is a radically
different language than C, an SML binding will test the
“interfaceability” of Gtk+. This is an important design
goal for the Gtk+ developers.

Standard ML
SML is a functional language with imperative features
which is widely used for teaching and research. It is

roughly on the same level of abstraction as Python and
Scheme. In contrast to Python and Scheme, which are
dynamically typed, SML isstatically typed, like Java and
C++. This means that type errors are detected atcompile
time rather than atrun time. Despite static typing, it is
not necessary for the SML programmer to explicitly pro-
vide type annotations in the program. SML featurestype
inference: the compiler reconstructs type annotations as
needed.

SML is one of the few languages with a formal defi-
nition. The definition of SML [10] consists of 93 pages
of mathematical notation (a ”big step” structured opera-
tional semantics, plus type inference rules) and English
prose. The book is not meant as tutorial for the language.
Rather, it provides an implementation independent for-
mulation of SML. This formal definition means that it is
possible to write substantial applications in SML that are
not dependent on a specific compiler. There are also sev-
eral mature SML implementations with widely different
implementation strategies, ranging from byte-code in-
terpreters with interactiveread–eval–print–loopsto ag-
gressive whole-program optimizing native code compil-
ers.

Gtk+

Gtk+, The GIMP Toolkit [14], is an LGPL-licensed [6]
library for creating graphical user interfaces. It works on
many UNIX-like platforms, on Windows, and on Linux
framebuffer devices. Gtk+ is the graphical toolkit used
in the GNOME desktop environment. Gtk+ is imple-
mented in C, with an object-oriented hierarchy of user
interface elements (“widgets”). From the beginning, the
Gtk+ developers have paid attention to making it feasi-
ble and practical to develop “bindings” or “wrappers”
permitting Gtk+ use by programs written in program-
ming languages other than C. See [12] for a current list
of bindings.

The Gtk+ library itself only contains widgets, but it is
built on a number of useful libraries, with which it is of-
ten associated. Specific libraries used by Gtk+ are [13]:

GLib A general-purpose utility library, not specific to
graphical user interfaces. GLib provides many use-

ful data types, macros, type conversions, string util-
ities, file utilities, and so forth.

Pango A library for internationalized text handling. It
provides the rendering engine for the widgets in
Gtk+ that displays text.

ATK The Accessibility Toolkit. It provides a set of
generic interfaces allowing accessibility technolo-
gies to interact with a graphical user interface.
Gtk+ widgets have built-in support for accessibil-
ity using the ATK framework.

GDK The abstraction layer that allows GTK+ to sup-
port multiple windowing systems.

Our work, however, is concentrated on the widget set
itself.

Overview of this paper
The rest of this paper is organized as follows. Section 2
gives a brief introduction to SML. Section 3 shows a
“Hello World” example using mGTK. Section 4 explains
how a single-inheritance class-hierarchy, in particular
Gtk+’s, can be encoded in SML’s type system, while re-
taining type safety. Section 5 describes the way that the
mGTK library is built upon the mGTK infrastructure.
Section 6 describes practical matters of mGTK, such as
which SML compilers are supported. Finally, Section 7
lists related work and Section 8 gives some conclusions.

2 Brief Introduction to SML
This section gives a brief overview of some of the main
features of SML. It is not sufficient to serve as a stan-
dalone user programming guide for the language. How-
ever, it should be sufficient to get an understanding of
the examples in the rest of the paper. For more informa-
tion about SML, we refer the interested reader to one of
the many fine textbooks [8, 11] available.

SML is a two-level language. It consists of acore lan-
guagefor programming in the small (that is, functions,
data structures and algorithms), and amodule language
for programming in the large.

Figure 1 shows a small stack library implemented in
SML. This example shows most of the important fea-
tures of SML. The library consists of two parts: an in-
terface description, which is called asignaturein SML
(Figure 1(a)), and an implementation module, which is
called astructure in SML (Figure 1(b)). Informally
speaking, a signature is the “type” of a structure. It spec-
ifies the declarations of the structure that are to be exter-
nally visible.

The signature is namedSTACK. Its extent is delim-
ited bysig . . . end. It contains five specifications: one
type specification, one exception specification, and three
value specifications.

The type specificationtype ’a stack states the
constraints on a module that satisfies (implements) the

signatureSTACK. Such a module must declare a type
namedstack: this type is parameterized. The’a is a
type variable: Type variables can be instantiated to other
types. This is what is meant by a parameterized type.
Thus, the type of a stack of integers isint stack, the
type of a stack of integer stacks isint stack stack,
and so on. Type variables are at the core ofparamet-
ric polymorphism(similar to generics in, for example,
C++ and Java; see [7] for a comparison of programming
languages with support for parametric polymorphism).
Note that the type specification does not say anything
about how a stack must be implemented.

The exception specification states that an exception
namedEmptyStack must be declared.

The first value specification says that a constant
named empty must be declared and that this con-
stant must have type’a stack. That is, empty
is a polymorphic value: it can be used in contexts
where anint stack is needed or in contexts where a
int stack stack is needed. The next value specifica-
tion states that a function namedpush must be imple-
mented. This function takes two arguments, an element
and a stack, and returns a stack. Again, we see how type
variables are used to specify thatpush must work with
stacks, whereas the elements can have any type. The last
value specification states that a function namedpopmust
be implemented, and thatpop takes a stack as argument
and returns an element and a new stack.

Figure 1(b) shows the code of the implementation,
in the form of a structure declaration. The declarations
states that the structure is namedStack, thatStack sat-
isfies the signatureSTACK, and thatStack does not re-
veal any implementation details not revealed bySTACK
(the latter connoted by:>). The extent of a structure is
delimited bystruct . . . end.

The parameterized typestack is implemented by an
algebraic data type described by adatatype declara-
tion. This declaration says that an’a stack is either
the constantEmpty, or is built by applying the construc-
torStack to an element and a stack. (Constants declared
by adatatype declaration such asEmpty are known as
constructors).

The exception declarationexception EmptyStack
declares an exception.

The next declaration states thatempty is bound to
Empty. The functionpush just applies the constructor
Stack to its arguments. The functionpop is more in-
teresting. This function takes a stack as argument and
then uses acase expression to analyze its argument.
(Here we have reused the namestack. Types and val-
ues uses different name spaces: thus, the same name can
be used for both a type and a value.) If the argument
is the empty stack (the constantEmpty) then the excep-
tion EmptyStack is raised (thrown). Otherwise, the ar-



signature STACK = sig
type ’a stack
exception EmptyStack
val empty : ’a stack
val push : ’a * ’a stack -> ’a stack
val pop : ’a stack -> ’a * ’a stack

end

(a) interface

structure Stack :> STACK =
struct
datatype ’a stack =

Empty
| Stack of ’a * ’a stack

exception EmptyStack
val empty = Empty
fun push(elem, stack) =

Stack(elem, stack)
fun pop stack =

case stack of
Empty => raise EmptyStack

| Stack(top, rest) =>
(top, rest)

end

(b) implementation

Figure 1: Simple stack library implemented in SML.

gument has been constructed by applyingStack to the
argumentstop andrest, and then a pair consisting of
top andrest is returned.

Users of this library can call functions from the
structureStack by using “dot-notation”: for example,
Stack.pop mystack.

This small example illustrates one of the cornerstones
in functional programming: new values are constructed
by analyzing, composing, and sharing old values. This
is in contrast to imperative and object-oriented program-
ming, where values are copied and modified. (A new
trend in object-oriented programming is to simulate a
functional style. See for example [2, Item 13 and 14].)

3 “Hello World” in mGTK

Figure 2 shows a deliberately simple “Hello World” ex-
ample using mGTK. It illustrates (1) how to get the
toolkit initialized usingGtkBasis.init (from a mod-
ule containing basic Gtk+ functionality not related to
specific widgets), (2) how to construct new widgets (us-
ing moduleWindow for the Window widget, andButton
for the Button widget), and (3) how to connect signals to
widgets (using moduleSignal).

Even this small example shows some of the main ad-
vantages of combining SML with Gtk+. There are no
type annotations in the program source. Nonetheless, the
program is statically type-checked by the compiler: type
errors are found and reported at compile time rather than
at runtime. In the figure we use a SML construct not ex-
plained earlier: the expressionfn _ => false denotes
ananonymous functionthat returnsfalse regardless of
what argument is given (you can use thewildcard pat-

tern_ (underscore) to ignore an argument to a function).
Anonymous functions are often handy for simple call-
backs, such as this one.

The constructlet val x = expdeclares the identifier
x to be bound to the value obtained by evaluating the ex-
pressionexp. If the only reason for evaluatingexpis for
its side effect, one can use the wildcard pattern_ instead
of x. Expressions evaluated only for their side effects can
also be sequentialized using;. The value(), the nullary
tuple of typeunit, can be used as the return value of
purely side-effecting functions. Such syntacic conve-
niences improve the readability and quality of code.

Finally, in SML, the double-colon:: denotes the cons
operation on lists. That is, to add an elementx to the be-
ginning of a listxs we write x::xs (in contrast to C++
where double-colon is the module operator). Built-in
data types such as lists make GUI programming more
convenient.

4 Encoding of Classes

As described in Section 1, SML is a functional language
without object-oriented features, while Gtk+ is designed
as an object-oriented library. This mismatch makes con-
structing an SML interface to Gtk+ tough . The most
difficult problem is how to represent the subtype rela-
tions defined by a class hierarchy in SML’s type system.
In this section, we discuss how to present a type-safe
SML interface to the Gtk+ class hierarchy. Bytype-safe
we mean that when an SML application programmer us-
ing our library makes a type-error in calling into Gtk+
(calling a undefined method on object, for instance) the
SML compiler should give a type error at compile time.



structure HelloWorld = struct
fun hello _ = print "Hello World\n"

fun main _ =
let val _ = GtkBasis.init(CommandLine.name()::CommandLine.arguments())

val window = Window.new ()
val button = Button.new_with_label "Hello World"

in Signal.connect window (Widget.delete_event_sig (fn _ => false))
; Signal.connect window (Widget.destroy_sig GtkBasis.main_quit)
; Signal.connect button (Button.clicked_sig hello)
; Container.add window button
; Widget.show_all window
; GtkBasis.main()

end
end

val _ = HelloWorld.main()

Figure 2: Hello World in mGTK.

Throughout, we shall think of class hierarchies mainly
as definitions of subtype relationships. This “confusion”
of classes and types is intentional: it is standard prac-
tice in types for object oriented programming. We are
able to take advantage of two properties of Gtk+ and
SML. First, Gtk+ implements only a single-inheritance
class system. Second, SML’s type system is expres-
sive enough to express the subtype relations of single-
inheritance class hierarchies.

We present a general method of taking a given ob-
ject oriented class hierarchy and encoding it in the SML
type system. The properties of the resulting encoding
are: each class type has a corresponding SML type; the
encoding iscomplete(all typings allowed by the class hi-
erarchy is also allowed by the encoding); and the encod-
ing is sound(all typings that is disallowed by the class
hierarchy is also disallowed by the encoding). The last
property is also called “type safety”. In type-theoretic
jargon, the trick is to use parametric polymorphism and
existential typesto encode inheritance subtyping. In par-
ticular we usephantom typesto encode the inheritance
path.

Figure 3 shows a small class hierarchy with just four
classes.Label andContainer are subclasses of the class
Widget andWindow is a subclass ofContainer. For the
specific class hierarchy in Figure 3, the properties we
want to enforce with our type encoding into SML’s type
system are: that we should be allowed to call theshow
method on all kinds of arguments whether they are of
class typeWidget, Label, Container, or Window; that
we should be allowed to call theadd method onCon-
tainers andWindows, but notWidgets andLabels; that
we should only be allowed to callset_title on Win-
dows; and so on.

Widget

+show(): void

Label

+set_text(text:string)

Container

+add(w:widget): void

Window

+set_title(title:string): void

Figure 3: Small example class hierarchy

We now describe the details of the encoding of a class.
Throughout this description we only present the SML
specifications, that is, the parts that go into the signa-
tures. The parts that go into the structures are not as
interesting: that is simply a matter of calling into the C
runtime.

Class types: A base class likeWidget in Figure 3 is en-
coded as an abstract parameterized type:

type ’path widget

(We follow the convention suggested by the Stan-
dard ML Basis Library and spell type-names in lower-
case, with underscores if needed.) The type variable
’path will be used to hold the inheritance path for sub-
classes.

Subtyping/Inheritance: For a subclass likeLabel we
need to encode two things: the existence of the class
(type), and the subtype relation to the parent class. To
do this, we declare two new SML types: an abstract pa-



rameterized type, and a type abbreviation for specifying
the inheritance.

type ’path label_t
type ’path label =

’path label_t widget

We call the abstract type (herelabel_t) the witness
typebecause it witnesses that the class exists. Similar,
the typelabel is the type abbreviation that specifies that
Label inherits fromWidget. In the declaration oflabel
we see that the type variable’path in the declaration
of the typewidget has been instantiated with the type
expression’path label_t, which contains a new type
variable (also named’path). In the rest of the paper we
shall use the convention that witness types ends with_t.

This is the juicy bit of the encoding, because this is re-
ally what makes it possible to encode single-inheritance
class hierarchies in SML. Unfortunately, this is also the
hardest part of the encoding to understand. Our experi-
ence is that you have to work a bit with some code to
really comprehend the trick.

Methods: Because SML is not an object-oriented lan-
guage we shall model methods with ordinary functions.
We use the usual convention that the first argument is
the object on which the method is called. (Gtk+ also
uses this convention.)

We can now write the type for the methodadd in Con-
tainer:

val add : ’path container -> ’a widget
-> unit

This specification says thatadd takes two arguments, an
object of typeContainer and a widget, and that it returns
unit as result. Similarly, the methodset_title from
classWindow has the type:

val set_title : ’path window -> string
-> unit

That is,set_title takes two arguments, an object of
typeWindow and a string, and it returnsunit.

Constructors: We have to be a bit careful with con-
structors. If we return a value with a polymorphic type-
variable’path that holds an inheritance path that has
not yet been “plugged”, then we could accidentally use
a super-class constructor to construct values that can be
instantiated to the type of a sub-class. Hence, we intro-
duce the abstract dummy typebase and use that to plug
the type variable. Thus, the type of the constructor for
Label is:

type base
val new : unit -> base label

The convention in Gtk+ is that constructors are named
new.

Fields: We are not able to handle fields directly, be-
cause we keep the representation of objects completely
opaque. Thus, all inspections of and changes to fields
must be done through accessor methods.

We then wrap all parts of the encoding of a class into a
signature/structure pair of its own. That is, for the class
Window in Figure 3 the SML signature is:

signature Window =
sig
type ’path window_t
type ’path window =

’path window_t Container.container
val new : unit -> GtkBasis.base window
val set_title : ’path window -> string

-> unit
end

We see that this signature relies on two structures:
Container for the classContainer, andGtkBasis for
the dummy typebase. In addition to the signature
Window we also need a structure calledWindow that im-
plements the actual calls to the relevant Gtk+ C func-
tions.

Does this encoding really allow all the things that the
Gtk+ class hierarchy allows? Yes. For example, the
functionContainer.add has type:

val add : ’p1 container -> ’p2 widget
-> unit

From this type we can see that, iflabel is
a value of type base label and window is a
value of type base window then the application
Container.add window label is well-typed be-
cause: (1) the type ofwindow is just a abbrevia-
tion for base window_t container, thus, the type
variable ’p1 can be instantiated tobase window_t,
and (2) the type oflabel is just an abbreviation of
base label_t widget, thus, the type variable’p2
can be instantiated tobase label_t.

Consider now the functionset_title that only
works onWindows:

val set_title : ’p window -> string -> unit

If we, by mistake, attempt to use this function to set the
text of the labellabel (with type base label) as in
expressionWindow.set_title label "New text",
we get a (compile-time) type error saying (essentially)
that theLabel widget is not a subclass of theWindow
widget because the inheritance paths do not match. Here
is the concrete error message given by the Moscow ML
compiler:



- Window.set_title label "New text";
! Toplevel input:
! Window.set_title label "New text";
! ^^^^^
! Type clash: expression of type
! base label_t widget
! cannot have type
! ’a window_t container_t widget

Hence, we have demonstrated that for these concrete
examples our encoding is both sound and complete.

5 Process
In constructing the mGTK binding we leverage the fore-
sightedness of the Gtk+ developers. Early on, they rec-
ognized that it would be important to have a machine-
readable “specification” of the toolkit. The specification
would describe the widget classes, the inheritance hier-
archy, and methods and functions in the toolkit. This
specification was implemented using a lisp-like custom
notation in thegtk.defs file of the toolkit. One could
argue that it is simple enough to extract the same infor-
mation from the C header files. However, C headers are
difficult to parse, whereas the defs format is straightfor-
ward to parse.

The bulk of the mGTK binding is constructed auto-
matically from thegtk.defs file. The complete binding
process is naturally divided into two phases: (1) bind-
ing design, where we apply the principles described in
Section 4 to a few representative widgets to demonstrate
the structure of the binding, and (2) binding construc-
tion, where the structure in (1) is applied to the entire
toolkit. It is important to note here that the design phase
can be carried out for a very small subset of the toolkit,
after which the construction phase “mimics” that for the
complete toolkit.

This phase separation makes it easier to get the design
right, simply because there are fewer issues to deal with.
It also makes the work involved in moving the binding to
other SML compilers manageable: the compiler writers
can provide the equivalent of the small subset for their
compiler, and utilize that style during the construction
phase. We also hope that the phase separation will help
when new releases of Gtk+ are produced. Most of the
work in constructing the binding for the new release is
over when the design of the small subset has been com-
pleted.

Let us return to our running example, and look
at some example specifications of widgets, func-
tions/methods, and signals. Figure 4 shows three entries
in the gtk.defs file. The first entry shows a widget
specification indicated bydefine-object. From the
entry we see that theGtkContainer widget (the name
appearing right afterdefine-object is a shorthand) in-

(define-object Container
(in-module "Gtk")
(parent "GtkWidget")
(c-name "GtkContainer")
(gtype-id "GTK_TYPE_CONTAINER")

)

(define-method gtk_container_add
(of-object "GtkContainer")
(c-name "gtk_container_add")
(return-type "none")
(parameters
’("GtkWidget*" "widget")

)
)

(define-signal delete-event
(of-object "GtkWidget")
(return-type "gboolean")
(when "last")
(parameters
’("GdkEventAny*" "p0")

)
)

Figure 4:gtk.defs excerpt.

herits fromGtkWidget, and it belongs in theGtk mod-
ule. We also see the type assigned to instances of this
widget in theGtk+ type system(which is completely un-
related to the SML encoding given above).

The next entry shows a method specification for the
methodadd. This method takes aGtkWidget* (in the C
implementation) argument, and returns nothing. Since it
is a method, there is an implicit “self” argument of type
GtkContainer*.

The final entry shows a “signal handler” or (“call-
back”) specification. In this case, we specify the proto-
type for handlers of delete events on widgets. The signal
handler fordelete-event for widgetGtkWidgets ac-
cepts a parameter of typeGdkEventAny*, and returns a
value of typegboolean.

6 The mGTK Binding

The mGTK binding is available at SourceForge
http://mgtk.sf.net/ and is released under the GNU
Lesser General Public License (LGPL) [6].

A fundamental difference in producing SML bindings
of Gtk+, compared to bindings for other languages, is
the existence of a variety of compilers (Section 1). This
sets this work apart from bindings to languages such as
Python, where there is only one target compiler and run-
time system.



The encoding of the Gtk+ class hierarchy in the SML
type system in Section 4 isthecore aspect of the bind-
ing. As the encoding stays within the language as de-
fined in the Definition [10], this aspect of the binding
remains the same for all SML compilers conforming to
the Definition. In other words, the interface exposed to
the application programmer is the same across all com-
pilers. One finds SML and Gtk+ implementations on a
large variety of platforms. Thus, the GUI porting work
in moving application programs from one of these plat-
forms to another is largely eliminated.

The mGTK binding already targets two of the main
SML systems, Moscow ML [17] and MLton [16]. The
authors are currently looking into constructing bindings
for other SML compilers (in particular, theML Kit with
Regions[15] andSML.NET[18] with Gtk#). As men-
tioned earlier (Section 5), the issues here mainly involve
interfacing to C.

The potential for partial compiler independence sets
the present binding apart from other Gtk+ bindings for
SML; notably, theSML-Gtk binding for the SML of
New Jersey compiler [9]. TheSML-Gtk binding is
also based on phantom types. Our binding predates
theSML-Gtk binding by approximately two years—the
SML-GtkUser’s Manual refers to the mGTK binding. To
date no serious attempts has been made to merge these
two projects. The reason for this is that, even though the
projects seems similar, we have followed rather differ-
ent strategies for constructing our respective bindings.
SML-Gtk is partly generated by theml-nlffi foreign
function interface, for instance, and does not attempt to
automate memory management.

7 Related Work
The list of language bindings for Gtk+ shows a plethora
of different languages from which Gtk+ is accessible. In
this section we briefly discuss the bindings most related
to mGTK.

When considering ML-like languages, there are two
major alternatives to the mGTK binding. TheSML-Gtk
binding was discussed earlier. Thelablgtk binding is
a Gtk+ binding for O’Caml. O’Caml is a ML dialect
different from SML: among other things, it has object-
oriented features. This binding, therefore, can directly
utilize the Gtk+ object hierarchy.

Gtk+ has also been bound to other functional lan-
guages. For example,gtk+hs is a Haskell binding, and
erlgtk is an Erlang binding. Bindings also exist for
other graphical toolkits. For example,sml_tk is an
SML binding of Tk.

The use of phantom types to express invariants about
programs is not new. However, the encoding of a single-
inheritance hierarchy as above is original with us. Inde-
pendent work has established similar results [5]. On the

construction side of things, other bindings are also ma-
chine generated. For this, some of the bindings use the
Simplified Wrapper and Interface Generator(SWIG)
[1], while others extract appropriate information directly
from the C headers files of Gtk+.

From the outset, the necessity of access to libraries
has been realized in the functional programming com-
munity. Work in this area for SML includes SML/NJ’s
foreign function interface [3]; for Haskell it includes [4].

8 Conclusions and Future Work
It is our intention to continue this work by utilizing ap-
propriate programming language technology to gradu-
ally bind more and more of the GNOME development
platform for SML. As was the case above, this entails
designing appropriate representations of the platform in
the SML world (in particular, preserving the type-safety
property mentioned above). It also includes the more
practical work of extending the code generator to handle
such newly introduced representations.

The long term goal for mGTK is to target most
of the GNOME platform. The advantages of bring-
ing GNOME to the SML community in the form of
such bindings are twofold. Firstly, it would allow
SML programmers access to the vast collection of use-
ful application-level support in GNOME. Secondly, it
would allow SML programmers to take part in the de-
velopment of GNOME components, by allowing them
to write such components in SML. The key technical as-
pect to be solved here is to support type-safe inheritance
on the SML side of things. Of course, one will also have
to explore exactly how to tie the various languages to-
gether. The GNOME community already has experience
in this area.

In this paper we have demonstrated that it is theoret-
ically and practically possible to make a type-safe in-
terface from SML to Gtk+. This is interesting for sev-
eral reasons. First, mGTK was one of the first graphical
toolkits available to the SML community. Second, the
fact that it is possible to make an SML binding to Gtk+
attests to the claimed “interfaceability” of Gtk+, because
SML is so radically different from C in abstraction level
and paradigm. Third, by auto-generating the binding, we
get a binding of the complete Gtk+ toolkit. Finally, we
believe that the particular way we construct the binding
can be extended to bind the entire GNOME development
platform, using mainly machine generated stub code.

